Einführung in GAP
am Beispiel der Rubik's Groups

Olaf Encke   Viktoria Idt

Universität Ulm

11. Dezember 2002

Gliederung

Rubik's Cube und Rubik's Group

Es sei $\mathcal{G}$ Rubik's Group, also die von den Seitendrehungen von Rubik's Cube erzeugte Gruppe, die die $48 = 6 \cdot 8$ farbigen Quadrate permutiert (wobei die Orientierungen der Seitenmitten vernachlässigt werden).

```
                     +--------------+
                     | 1    2    3  |
                     |              |
                     | 4   up    5  |
                     |              |
                     | 6    7    8  |
                     +--------------+
                     |01  02  03|
                     |04   U  05|
                     |06  07  08|
          +--  --  --+--  --  --+--  --  --+--  --  --+
          |09  10  11|17  18  19|25  26  27|33  34  35|
          |12   L  13|20   F  21|28   R  29|36   B  37|
          |14  15  16|22  23  24|30  31  32|38  39  40|
          +--  --  --+--  --  --+--  --  --+--  --  --+
                     |41  42  43|
                     |44   D  45|
                     |46  47  48|
                     +--  --  --+
```

Erzeuger

Wir entnehmen diesem Modell die Erzeuger und erstellen mit diesen die Rubik's Group $\mathcal{G}$ in GAP.

```
gap> cube:=Group((1,3,8,6)(2,5,7,4)(9,33,25,17)(10,34,26,18)(11,35,27,19),
>    (1,17,41,40)(4,20,44,37)(6,22,46,35)(9,11,16,14)(10,13,15,12),
>    (6,25,43,16)(7,28,42,13)(8,30,41,11)(17,19,24,22)(18,21,23,20),
>    (3,38,43,19)(5,36,45,21)(8,33,48,24)(25,27,32,30)(26,29,31,28),
>    (1,14,48,27)(2,12,47,29)(3,9,46,32)(33,35,40,38)(34,37,39,36),
>    (14,22,30,38)(15,23,31,39)(16,24,32,40)(41,43,48,46)(42,45,47,44));
<permutation group with 6 generators>
```

Zur Kontrolle lassen wir uns die Erzeuger der Gruppe wieder ausgeben.

```
gap> GeneratorsOfGroup(cube);
[ (1,3,8,6)(2,5,7,4)(9,33,25,17)(10,34,26,18)(11,35,27,19),
  (1,17,41,40)(4,20,44,37)(6,22,46,35)(9,11,16,14)(10,13,15,12),
  (6,25,43,16)(7,28,42,13)(8,30,41,11)(17,19,24,22)(18,21,23,20),
  (3,38,43,19)(5,36,45,21)(8,33,48,24)(25,27,32,30)(26,29,31,28),
  (1,14,48,27)(2,12,47,29)(3,9,46,32)(33,35,40,38)(34,37,39,36),
  (14,22,30,38)(15,23,31,39)(16,24,32,40)(41,43,48,46)(42,45,47,44) ]
```

Ordnung und Bahnen

Nun lassen wir uns die Ordnung der Gruppe $\mathcal{G}$ ausgeben.

```
gap> Size(cube);
43252003274489856000

gap> Collected(Factors(last));
[ [ 2, 27 ], [ 3, 14 ], [ 5, 3 ], [ 7, 2 ], [ 11, 1 ] ]
```

Die Ordnung von $\mathcal{G}$ beträgt also $2^{27} \cdot 3^{14} \cdot 5^3 \cdot 7^2 \cdot 11^1$ und liegt in der Größenordnung von 10^{19}.
Auch die Bahnen von $\mathcal{G}$ bringen wir schnell in Erfahrung.

```
gap> Orbits(cube);
[ [ 1, 3, 17, 14, 8, 38, 9, 41, 19, 48, 22, 6,
30, 33, 43, 11, 46, 40, 24, 27, 25, 35, 16, 32 ],
  [ 2, 5, 12, 7, 36, 10, 47, 4, 28, 45, 34, 13,
29, 44, 20, 42, 26, 21, 37, 15, 31, 18, 23, 39 ] ]
```

Bild der Operation auf den Eckwürfeln

Die Operation von $\mathcal{G}$ auf den 8 Eckwürfeln liefert einen Gruppenhomomorphismus $\mathcal{G} \longrightarrow C_3^8 \rtimes S_8$. Wir bestimmen nun dessen Bild.

```
gap> o1:=Orbit(cube,1);
[ 1, 6, 40, 27, 8, 35, 16, 41, 32, 25, 48, 3,
11, 24, 46, 33, 43, 17, 30, 14, 19, 9, 22, 38 ]
```

Wir erstellen eine neue Gruppe, die nur die Ecken des Rubik's Cube berücksichtigt.

```
gap>cube_corners:=Group((1,3,8,6)(9,33,25,17)(11,35,27,19),
(1,17,41,40)(6,22,46,35)(9,11,16,14),
(6,25,43,16)(8,30,41,11)(17,19,24,22),
(3,38,43,19)(8,33,48,24)(25,27,32,30),
(1,14,48,27)(3,9,46,32)(33,35,40,38),
(14,22,30,38)(16,24,32,40)(41,43,48,46));
<permutation group with 6 generators>
gap> Size(cube_corners);
88179840
```

Bild der Operation auf den Eckwürfeln

Wir untersuchen, ob wir die Elemente der Gruppe zusammenfassen können, ob die Bewegungen einzelner Elemente voneinander abhängig sind.

```
gap> corners:=Blocks(cube_corners, o1);
[ [ 1, 9, 35 ], [ 3, 27, 33 ], [ 6, 11, 17 ], [ 8, 19, 25 ],
[ 14, 40, 46 ], [ 16, 22, 41 ], [ 24, 30, 43 ], [ 32, 38, 48 ] ]
```

$\Rightarrow$ d.h. wenn z.B. 6 auf 24 geht, so geht 11 auf 30 bzw. 43.
Nun untersuchen wir die Operation der Gruppe auf die acht Ecken. Wir wissen, dass die Gruppe `cube_corners` auf eine Untergruppe von S_8 abgebildet wird (welche Untergruppe es ist, ist noch nicht bekannt).

```
gap> blockhom1:=ActionHomomorphism(cube_corners,corners,OnSets);
<action homomorphism>

gap> cube_corners_i:=Image(blockhom1);
Group([(1,2,4,3),(1,3,6,5),(1,5,8,2),(3,4,7,6),(5,6,7,8),(2,8,7,4)])
           F         L         U         D         B         R
gap> Size(cube_corners_i);
40320
```

Bild der Operation auf den Eckwürfeln

```
          +-------+
          |05   08|
          |   U   |
          |01   02|
  +-------+-------+-------+-------+
  |05   01|01   02|02   08|08   05|
  |   L   |   F   |   R   |   B   |
  |06   03|03   04|04   07|07   06|
  +-------+-------+-------+-------+
          |03   04|
          |   D   |
          |06   07|
          +-------+
```

Dieses entspricht Homomorphismus `blockhom1` : `cube_corners` $\longrightarrow S_8$, da eine Permutationsgruppe 8. Grades, welche Ordnung 40320 hat, die volle Symmetrische Gruppe S_8 sein muss.

Bild der Operation auf den Eckwürfeln

Als nächstes bestimmen wir den Kern des Homomorphismus `blockhom1`, eine Untergruppe von `cube_corners`, so dass die Position der Ecken gleich bleibt, diese aber eventuell noch gedreht werden können

```
gap> Kern1:=Kernel(blockhom1);
<permutation group of size 2187 with 7 generators>
```

Die Ordnung des Kerns entspricht $3^7 = 2187$. Der Kern ist der Normalteiler der Gruppe `cube_corners`, also suchen wir nun eine Untergruppe von `cube_corners`, deren semidirektes Produkt mit dem Kern die gesamte Gruppe `cube_corners` ergibt.

```
gap> Compclass1:=Complementclasses(cube_corners,Kern1);
[ <permutation group with 3 generators> ]
gap> Compclass1[1];
<permutation group with 3 generators>
gap> GeneratorsOfGroup(Compclass1[1]);
[ (1,6,8,3)(9,17,25,33)(11,19,27,35),
  (1,3,6,8,14,16,32)(9,33,17,25,46,22,38)(11,19,40,41,48,35,27),
  (1,3,6,8,14,24,32)(9,33,17,25,46,30,38)(11,19,40,43,48,35,27) ]
gap> Size(Compclass1[1]);
40320
```

Bild der Operation auf den Eckwürfeln

Wir zeigen nun, dass die gefundene Gruppe tatsächlich die gesuchte Untergruppe ist, d.h.

1. der Durchschnitt zwischen dem Kern und einer Komplementklasse nur die 1 enthält:

```
gap>Size( Intersection ( Compclass1[1],Kern1));
1
```

2. das semidirekte Produkt der Komplementklasse und dem Kern die gesamte Gruppe cube_corners ergibt:

```
gap> ClosureGroup(Compclass1[1],Kern1)=cube_corners;
true
```

Das Ergebniss unserer Untersuchung ist, dass das Bild des Gruppenhomomorphismus von $\mathcal{G} \longrightarrow C_3^8 \rtimes S_8$ das semidirekte Produkt von $C_3^7 \rtimes S_8$ ist.

Bild der Operation auf den Kantenwürfeln

Die Operation von $\mathcal{G}$ auf den 12 Kantenwürfeln liefert einen Gruppenhomomorphismus $\mathcal{G} \longrightarrow C_2^{12} \rtimes S_{12}$. Auch hier bestimmen wir analog zu den Eckwürfeln das Bild.

```
gap> o2:=Orbit(cube,2);
[ 2, 5, 12, 7, 36, 10, 47, 4, 28, 45, 34, 13, 29,
 44, 20, 42, 26, 21, 37, 15, 31, 18, 23, 39 ]

...

gap> GeneratorsOfGroup(cube_edges);
[ (2,5,7,4)(10,34,26,18),
  (4,20,44,37)(10,13,15,12),
  (7,28,42,13)(18,21,23,20),
  (5,36,45,21)(26,29,31,28),
  (2,12,47,29)(34,37,39,36),
  (15,23,31,39)(42,45,47,44) ]

gap> Size(cube_edges);
980995276800
```

Bild der Operation auf den Kantenwürfeln

```
gap> edges:=Blocks(cube_edges,o2);
[ [ 2, 34 ], [ 4, 10 ], [ 5, 26 ], [ 7, 18 ], [ 12, 37 ], [ 13, 20 ],
  [ 15, 44 ], [ 21, 28 ], [ 23, 42 ], [ 29, 36 ], [ 31, 45 ], [ 39, 47 ] ]

gap> blockhom2:=ActionHomomorphism(cube_edges,edges,OnSets);
<action homomorphism>

gap> cube_edges_i:=Image(blockhom2);
Group([(1,3,4,2), (1,5,12,10), (2,6,7,5), (3,10,11,8), (4,8,9,6), (7,9,11,12)])
gap> Size(cube_edges_i);
479001600
```

```
              +------+
              |  12  |
              |05  10|
              |  01  |
       +------+------+------+------+
       |  05  |  01  |  10  |  12  |
       |07  02|02  03|03  11|11  07|
       |  06  |  04  |  08  |  09  |
       +------+------+------+------+
              |  04  |
              |06  08|
              |  09  |
              +------+
```

Bild der Operation auf den Kantenwürfeln

```
gap> Kern2:=Kernel(blockhom2);
<permutation group of size 2048 with 11 generators>
gap> Compclass2:=Complementclasses(cube_edges,Kern2);
[ <permutation group with 3 generators>, <permutation group with 3 generators>,
  <permutation group with 3 generators>, <permutation group with 3 generators>]
gap> GeneratorsOfGroup(Compclass2[1]);
[ (2,4,7,5)(10,18,26,34),
  (2,4,5,7,12,13,15,21,23,29,39)(10,26,18,37,20,44,28,42,36,47,34),
  (2,4,5,7,12,13,15,21,23,31,39)(10,26,18,37,20,44,28,42,45,47,34) ]
gap> GeneratorsOfGroup(Compclass2[2]);
[ (2,10,7,26)(4,18,5,34)(12,37)(13,20)(15,44)(21,28)(23,42)(29,36)(31,45)
     (39,47),
  (2,4,5,7,12,13,15,21,23,29,39)(10,26,18,37,20,44,28,42,36,47,34),
  (2,4,5,7,12,13,15,21,23,31,39)(10,26,18,37,20,44,28,42,45,47,34) ]
gap> Size(Compclass2[1]);
479001600

gap> Size(Intersection(Compclass2[1],Kern2));
1
gap> ClosureGroup(Compclass2[1],Kern2)=cube_edges;
true
```

Als Ergebnis der Untersuchung erhalten wir, dass das Bild des Gruppenhomomorphismus von $\mathcal{G} \longrightarrow (C_2)^{12} \rtimes S_{12}$ das semidirekte Produkt von $(C_2)^{11} \rtimes S_{12}$ ist.

Gezielte Manipulation des Rubik's Cube

Wir wollen nun einige typische Fragestellungen beantworten, die bei der genaueren Betrachtung des Rubik's Cube auftreten.

- Kann man eine Ecke alleine im Uhrzeigersinn drehen? (alles andere soll fest bleiben)

```
gap> (6,11,17) in cube;
false
gap> (6,17,11) in cube;
false
```

Gruppentheoretisch kann man auch argumentieren, dass, wenn man eine Ecke alleine drehen könnte, man $(C_3)^8$ bekommen würde. Da wir allerdings nur $(C_3)^7$ haben, sind die Drehungen der Ecken nicht vollkommen unabhängig voneinander.

Gezielte Manipulation des Rubik's Cube

- Wie kann man 2 beliebige Eckwürfel drehen?
  - nächstliegende Ecke

```
gap> (6,11,17)(8,19,25) in cube;
false
gap> (6,17,11)(8,19,25) in cube;
true
```

  - diagonale Ecke

```
gap> (6,11,17)(24,30,43) in cube;
true
gap> (6,17,11)(24,30,43) in cube;
false
```

  - gegenüberliegende Ecke

```
gap> (6,11,17)(32,38,48) in cube;
true
gap> (6,17,11)(32,38,48) in cube;
false
```

```
                +--------+
                |01 02 03|
                |04  U 05|
                |06 07 08|
+--------+--------+--------+--------+
|09 10 11|17 18 19|25 26 27|33 34 35|
|12  L 13|20  F 21|28  R 29|36  B 37|
|14 15 16|22 23 24|30 31 32|38 39 40|
+--------+--------+--------+--------+
                |41 42 43|
                |44  D 45|
                |46 47 48|
                +--------+
```

Gezielte Manipulation des Rubik's Cube

- Kann man einen Kantenwürfel alleine kippen?
  2 Kantenwürfel?

```
gap> (7,18) in cube;
false

gap> (7,18)(21,28) in cube;
true
gap> (7,18)(23,42) in cube;
true
gap> (7,18)(31,45) in cube;
true
gap> (7,18)(39,47) in cube;
true
```

```
              +--  --  --+
              |01  02  03|
              |04   U  05|
              |06  07  08|
+--  --  --+--  --  --+--  --  --+--  --  --+
|09  10  11|17  18  19|25  26  27|33  34  35|
|12   L  13|20   F  21|28   R  29|36   B  37|
|14  15  16|22  23  24|30  31  32|38  39  40|
+--  --  --+--  --  --+--  --  --+--  --  --+
              |41  42  43|
              |44   D  45|
              |46  47  48|
              +--  --  --+
```

Isomorphiebeweis

Wir wollen zeigen, dass $\mathcal{G}$ isomorph ist zu einer Untergruppe vom Index 2 von $(C_2^{11} \rtimes S_{12}) \times (C_3^7 \rtimes S_8)$.

```
gap> Size(cube_corners);
88179840
gap> Size(cube_edges);
980995276800
gap> Size(cube_corners)*Size(cube_edges)/2;
43252003274489856000
gap> Size(cube);
43252003274489856000
```

Rubik's Extended Group

Es sei Γ Rubik's Extended Group, das ist Rubik's Group, bei der man die Orientierungen der Seitenmitten berücksichtigt (z.B. indem man Pfeile auf die Seitenmitten klebt).
Die Erzeuger von Γ als Permutationen auf 72 Punkten lauten:

```
gap> GeneratorsOfGroup(cubexl);
[ (1,3,8,6)(2,5,7,4)(9,33,25,17)(10,34,26,18)(11,35,27,19)(49,50,51,52),
  (1,17,41,40)(4,20,44,37)(6,22,46,35)(9,11,16,14)(10,13,15,12)(53,54,55,56),
  (6,25,43,16)(7,28,42,13)(8,30,41,11)(17,19,24,22)(18,21,23,20)(57,58,59,60),
  (3,38,43,19)(5,36,45,21)(8,33,48,24)(25,27,32,30)(26,29,31,28)(61,62,63,64),
  (1,14,48,27)(2,12,47,29)(3,9,46,32)(33,35,40,38)(34,37,39,36)(65,66,67,68),
  (14,22,30,38)(15,23,31,39)(16,24,32,40)(41,43,48,46)(42,45,47,44)
       (69,70,71,72) ]

gap> Size(cubexl);
88580102706155225088000
gap> Collected(Factors(last));
[ [ 2, 38 ], [ 3, 14 ], [ 5, 3 ], [ 7, 2 ], [ 11, 1 ] ]
```

Die Ordnung von Γ beträgt somit $2^{38} \cdot 3^{14} \cdot 5^{3} \cdot 7^{2} \cdot 11^{1} \approx 8,8 \cdot 10^{22}$.

Isotropiegruppe/Stabilisator

Wir suchen nun die Untergruppe $\mathcal{D}$ von Γ, die nur die Seitenmitten dreht, den übrigen Würfel aber fest lässt.

```
gap> ugr_d:=Stabilizer(cubexl, [1 .. 48], OnTuples);
<permutation group of size 2048 with 6 generators>
gap> IsSubgroup(cubexl,ugr_d);
true
gap> GeneratorsOfGroup(ugr_d);
[ (69,71)(70,72),(65,66,67,68)(69,72,71,70),(61,62,63,64)(65,66,67,68),
  (53,54,55,56)(57,59)(58,60)(65,66,67,68),
  (57,58,59,60)(61,64,63,62)(65,68,67,66)(69,70,71,72),
  (49,50,51,52)(53,56,55,54)(57,58,59,60)(61,64,63,62)(65,66,67,68)
       (69,72,71,70) ]
```

Normalteiler von Γ

Als letztes zeigen wir, dass $\mathcal{D} \leq Z(\Gamma)$ ein zentraler Normalteiler von Γ ist und $\Gamma/\mathcal{D} \cong \mathcal{G}$.

```
gap> cubexl_c:=Center(cubexl);
<permutation group of size 4096 with 9 generators>
gap> IsSubgroup(cubexl_c, ugr_d);
true
gap> IsNormal(cubexl, ugr_d);
true

gap> Size(cubexl)/Size(ugr_d);
43252003274489856000
gap> Size(cube);
43252003274489856000
```

Interessante Links

- http://www.rubiks.com
- http://lar5.com/cube/
- http://www.ssie.binghamton.edu/fridrich/cube.html
- http://www.twistymegasite.com
- http://web.usna.navy.mil/~wdj/rubik.html
- http://www.speedcubing.com
- http://home.t-online.de/home/kociemba/cube.htm
- http://www.cubeman.org

Ende

Vielen Dank für Ihre Aufmerksamkeit!

olaf.encke@gmx.de
viktoria.idt@gmx.de

created with LaTeX and beamer